
Faktory Worker
A worker client for Faktory.
Documentation
Faktory Worker documentation is available at https://hexdocs.pm/faktory_worker.
Getting up and running
To get started with Faktory Worker first add the dependency to your mix.exs file.
defp deps do
  [
    {:faktory_worker, "~> 1.2.0"}
  ]
end
Faktory Worker can then be configured to start as part of your application by adding it to the Application.start/2 function.
defmodule MyApp.Application do
  use Application

  def start(_, _) do
    children = [
      FaktoryWorker
    ]

    Supervisor.start_link(children,
      strategy: :one_for_one,
      name: MyApp.Supervisor
    )
  end
end
Make sure you have configured your application to start in your mix.exs file.
def application do
  [
    mod: {MyApp.Application, []}
  ]
end
This will start Faktory Worker using the default configuration. For details on how to configure Faktory Worker
see the FaktoryWorker documentation.
Next you can create a worker to handle job processing.
defmodule MyApp.SomeWorker do
  use FaktoryWorker.Job

  def perform(job_data) do
    do_some_work(job_data)
  end
end
Now you can start sending jobs to faktory and they will be automatically picked up by the worker.
:ok = MyApp.SomeWorker.perform_async("hey there!")
Sending multiple job arguments
It's possible to send more than one argument to Faktory by passing a list of data to perform_async/1. Picking up from the example in the Getting up and running section we can modify our worker by adding another perform function with an arity that matches the number of job arguments we expect to send to Faktory.
defmodule MyApp.SomeWorker do
  use FaktoryWorker.Job

  ...

  def perform(arg1, arg2) do
    do_some_work(arg1, arg2)
  end
end
With this new function in place you can now send multiple job arguments to Faktory.
:ok = MyApp.SomeWorker.perform_async(["arg 1", "arg 2"])
Configuration
The full list of configuration options are available in the Configuration documentation.
Logging
By default Faktory Worker will not output any log messages but instead emit events using the Telemetry library.
To enable the built in logging you will need to attach the default Telemetry handler provided by FaktoryWorker. The ideal place to do this is in your Application.start/2 callback.
defmodule MyApp.Application do
  use Application

  def start(_, _) do
    FaktoryWorker.attach_default_telemetry_handler()

    ...
  end
end
With this in place Faktory Worker will now output log messages for each of the events emitted.
For a full list of Faktory Worker events or for details on handling these events see the Logging documentation.
Contributing
We always appreciate contributions whether they are testing, reporting issues, feedback or submitting PRs. If you would like to work on Faktory Worker please follow the Developing section for details on how to get setup for developing and running the test suite.
Developing
Faktory Worker includes a docker compose file that provisions all of the Faktory instances required to run the test suite.
If you have docker compose installed you can run the up command from the Faktory Worker directory to start everything required.
$ docker-compose up -d
Creating faktory_worker_test          ... done
Creating faktory_worker_test_tls      ... done
Creating faktory_worker_password_test ... done

$ mix test
  
Faktory Worker Configuration
Available Configuration
Faktory Worker offers a range configuration options that are passed into the FaktoryWorker module directly when defining it as part of a supervision tree. All of these options are optional and have default values defined.
Configuration can be specified in the supervision tree using a two element tuple.
defmodule MyApp.Application do
  use Application

  def start(_, _) do
    children = [
      {FaktoryWorker, [name: :faktory_test]}
    ]

    ...
  end
end
Below is the full structure of available options and a definition for each one.
[
  name: atom(),
  connection: [
    host: String.t(),
    port: pos_integer(),
    password: String.t(),
    use_tls: boolean()
  ],
  pool: [
    size: pos_integer(),
    buffer_size: pos_integer() | :infinity
  ],
  worker_pool: [
    size: pos_integer(),
    disable_fetch: boolean(),
    queues: [
      list(String.t() | {String.t(), keyword()})
    ]
  ]
]
Option Definitions
	name (default: FaktoryWorker) - The name to use for this instance. This is useful if you want to run more than one instance of Faktory Worker. When using this make sure the workers are configured to use the correct instance (see Worker Configuration below).

	connection - A list of options to configure the connection to Faktory.
	host (default: "localhost") - The host name used to connect to Faktory.
	port (default: 7419) - The tcp port used to connect to Faktory.
	password (default: nil) - The password used to authenticate with Faktory.
	use_tls (default: false) - A boolean flag that indicates whether Faktory Worker should connect using a TLS connection.


	pool - A list of options to configure the pool of Faktory connections used for sending jobs to Faktory.
	size (default: 10) - The number of connections to open to Faktory.
	buffer_size (default: :infinity) - The number of jobs that can be stored in memory whilst waiting for an available connection from the pool. Note that this defaults to :infinity to prevent losing jobs during high load scenarios. This could potentially lead to increased memory consumption if there is low throughput to Faktory.


	worker_pool - A list of options to configure the Faktory workers.
	size (default: 10) - The number of worker connections to open to Faktory.
	disable_fetch (default: false) - A boolean flag used to disable fetching on each of the workers. This is useful if you want to prevent the workers from fetching new jobs without having to remove the queues config.
	queues (default: ["default"]) - A list of queues that the workers should fetch jobs from. See the Configuring Queues section below for more details.



Configuring Queues
Queues are configured by specifying a list of strings or tuples.
queues: ["queue_name", {"another_queue", opts}]
The queue name must match the name of the queue in Faktory. Using the tuple definition allows a list of options to be specified when fetching jobs from that queue.
The available options are.
[
  max_concurrency: pos_integer() | :infinity
]
Option Definitions
	max_concurrency (default: :infinity) - The maximum number of jobs that should be fetched and processed from this queue in parallel.

Worker Configuration
Workers can be configured with queue specific options that are used by Faktory Worker when sending jobs to Faktory. All of these options are optional and have default values.
Options are defined when using FaktoryWorker.Job.
use FaktoryWorker.Job, opts
Below is the full structure of available options and a definition for each one.
[
  queue: String.t(),
  retry: pos_integer(),
  reserve_for: post_integer(),
  custom: map(),
  faktory_name: atom()
]
Option Definitions
	queue (default: "default") - The name of the queue in Faktory to send jobs to.
	retry (relies on Faktory default) - The number of times the job should retry if it fails.
	reserve_for (relies on Faktory default) -The number of seconds the job is reserved for in Faktory before it is considered failed.
	custom (relies on Faktory default) - A map of values to be included with the job when it is sent to Faktory.
	faktory_name (default: FaktoryWorker) - The name of the Faktory instance to use when sending the job to Faktory.

  
Faktory Worker Logging
Faktory Worker uses the Telemetry library to emit events that can be handled by either Faktory Worker or the consuming application. This provides the flexibility for the consumer of Faktory Worker to choose how logging should behave.
By default Faktory Worker will not output any log messages. Logging needs to be explicitly enabled by calling the FaktoryWorker.attach_default_telemetry_handler/0 function.
Events can also be handled in the consuming application by manually attaching to Telemetry specifying the list of Faktory Worker events to handle. For instructions on creating a custom event handler see the Telemetry README.
All Faktory Worker events are prefixed with :faktory_worker and have a single event name. For example the full event name for the push event is [:faktory_worker, :push].
The measurements data included for all events will be a map containing a status field that will indicate the outcome of the event being emitted. The available outcomes are listed individually below for each type of event.
Push Event
Event Name: [:faktory_worker, :push].
The push event is emitted when a job has been sent to Faktory. The status of this event indicates whether the job was sent successfully.
The available status outcomes are.
	:ok - The job was successfully sent to Faktory.
	{:error, :not_unique} - The job was not sent to Faktory because another job with the same uniqueness already exists.

The metadata supplied with this event is a map containing the Faktory job arguments.
Fetch Event
Event Name: [:faktory_worker, :fetch].
The fetch event is emitted when attempting to fetch a job from Faktory. The status of this event indicates whether the fetch request was sent successfully.
The available status outcomes are.
	{:error, reason} - The request to fetch from Faktory failed. The reason will be any valid term indicating the reason the fetch failed.

The metadata supplied with this event is a map containing the following fields.
	wid - A worker identifier that can be used to identify which worker emitted this event.

Ack Event
Event Name: [:faktory_worker, :ack].
The ack event is emitted when reporting the success or failure of a processed job to Faktory. The status of this event indicates the outcome of processing the job.
The available status outcomes are.
	:ok - The job completed successfully and has been reported to Faktory.
	:error - The job failed to complete and has been reported to Faktory with details about the failure.

The metadata supplied with this event is a map containing the Faktory job arguments.
Failed Ack Event
Event Name: [:faktory_worker, :failed_ack].
The failed ack event is emitted when Faktory Worker was unable to report the outcome of processing a job to Faktory. The status of this event indicates the outcome of the processed job that failed to reach Faktory.
The available status outcomes are.
	:ok - The job completed successfully.
	:error - The job failed to complete.

The metadata supplied with this event is a map containing the Faktory job arguments.
Beat Event
Event Name: [:faktory_worker, :beat].
The beat event is emitted every time a worker has attempted to send a heartbeat message to Faktory. The status of this event indicates whether the heartbeat message was successfully received by Faktory.
The available status outcomes are.
	:ok - The heartbeat message was successfully received by Faktory.
	:error - The heartbeat message failed to reach Faktory.

The metadata supplied with this event is a map containing the following fields.
	wid - A worker identifier that can be used to identify which worker emitted this event.
	prev_status - The status that was emitted on the previous beat event. This is useful for tracking when the status has changed between heartbeats.

The FaktoryWorker module provides everything required to setup workers for sending and fetching jobs.
It is expected that FaktoryWorker will be configured and started as part of a supervision tree. Multiple instances of
FaktoryWorker can be configured by providing the :name  option which must be unique.
This module can either be configured using all default options.
children = [
  FaktoryWorker
]
Or by using the two element tuple format accepting a list of options as the second element.
children = [
  {FaktoryWorker, [name: :faktory_test, ...]}
]
For a full list of configuration options see the Configuration documentation.

    Attaches the default telemetry handler provided by FaktoryWorker.

      Specs

      
          attach_default_telemetry_handler() :: :ok | {:error, :already_exists}

      


Attaches the default telemetry handler provided by FaktoryWorker.
This function attaches the default telemetry handler provided by FaktoryWorker that
outputs log messages for each of the events emitted by FaktoryWorker.
For a full list of events see the Logging documentation.

The FaktoryWorker.Job module is used to perform jobs in the background by sending to and fetching from Faktory.
To build a worker you must use the job module within a module in your application.
defmodule MyApp.SomeWorker do
  use FaktoryWorker.Job
end
This will bring in all of the functionality required to perform jobs via Faktory. The MyApp.SomeWorker will now
have a perform_async/2 function available for sending jobs to Faktory. Before this function can be called, a
perform function must be defined. This function must accept the same number of arguments that are
being sent to Faktory via the perform_async/2 function.
defmodule MyApp.SomeWorker do
  use FaktoryWorker.Job

  def perform(job_arg) do
    do_some_work(job_arg)
  end
end
With this in place it is now possible to send work to Faktory and the MyApp.SomeWorker will fetch the job and call
the perform/1 function with the same job arguments that we sent.
> MyApp.SomeWorker.perform_async("job arg")
:ok
It is also possible to send multiple arguments for a single job by passing in a list of values to the perform_async/2
function.
> MyApp.SomeWorker.perform_async(["job arg1", "job arg2"])
:ok
In order for the job to be performed correctly, a perform/2 function needs to be defined within the MyApp.SomeWorker
module.
defmodule MyApp.SomeWorker do
  use FaktoryWorker.Job

  def perform(job_arg1, job_arg2) do
    do_some_work(job_arg1, job_arg2)
  end
end
When defining perform functions, they must always accept one argument for each item in the list of values passed into
perform_async/2.
Worker Configuration
A list of options can be specified when using the the FaktoryWorker.Job module. These options will be used when sending
jobs to faktory and will apply to all jobs sent with the perform_async/2 function.
For a full list of configuration options see the Worker Configuration documentation.
Overriding Worker Configuration
The perform_async/2 function accepts a keyword list as its second argument. This list has the same options
available that the FaktoryWorker.Job module accepts. Any options passed into this function override the options
that have been set on the worker module.
For a full list of configuration options see the Worker Configuration documentation.
Data Serialization
Faktory expects all values to be serialized in JSON format. FaktoryWorker uses Jason for serialization. This
means only values that implement the Jason.Encoder protocol are valid when calling the perform_async/2 function.

